# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python-flask-babelex
_pkgname=${pkgname#python-}
pkgver=0.9.4
pkgrel=10
pkgdesc='i18n and l10n support for Flask based on Babel and pytz.'
url='https://github.com/mrjoes/flask-babelex'
arch=('any')
license=('BSD')
depends=('python' 'python-flask' 'python-babel' 'python-speaklater'
         'python-jinja')
makedepends=('python-setuptools')
source=("https://github.com/mrjoes/flask-babelex/archive/$pkgver.tar.gz")
sha512sums=('e5591bcb83ca7e7644b1d06e604834369973bfba5050bd6890ad57125731981cda1e893544eea61cf69dca536e5473e2e50b08b36b0c3daee15d66f3815bd986')

build() {
  cd "$_pkgname-$pkgver"

  python setup.py build
}

package() {
  cd "$_pkgname-$pkgver"

  python setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build

  install -Dm 644 README -t "$pkgdir/usr/share/doc/$pkgname"
  install -Dm 644 LICENSE -t "$pkgdir/usr/share/licenses/$pkgname"
}

